content: -- Copyright (c) 2016-present, Facebook, Inc.
  -- All rights reserved.
  --
  -- This source code is licensed under the BSD-style license found in the
  -- LICENSE file in the root directory of this source tree.
  
  
  {-# LANGUAGE GADTs #-}
  {-# LANGUAGE LambdaCase #-}
  {-# LANGUAGE OverloadedStrings #-}
  
  module Duckling.AmountOfMoney.PT.Rules
    ( rules
    ) where
  
  import Data.Maybe
  import Data.String
  import Prelude
  
  import Duckling.AmountOfMoney.Helpers
  import Duckling.AmountOfMoney.Types (Currency(..), AmountOfMoneyData (..))
  import Duckling.Dimensions.Types
  import Duckling.Numeral.Helpers (isNatural, isPositive)
  import Duckling.Numeral.Types (NumeralData (..))
  import Duckling.Types
  import qualified Duckling.AmountOfMoney.Types as TAmountOfMoney
  import qualified Duckling.Numeral.Types as TNumeral
  
  ruleUnitAmount :: Rule
  ruleUnitAmount = Rule
    { name = "<unit> <amount>"
    , pattern =
      [ Predicate isCurrencyOnly
      , Predicate isPositive
      ]
    , prod = \case
        (Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.currency = c}:
         Token Numeral NumeralData{TNumeral.value = v}:
         _) -> Just . Token AmountOfMoney . withValue v $ currencyOnly c
        _ -> Nothing
    }
  
  ruleIntersectAndNumeral :: Rule
  ruleIntersectAndNumeral = Rule
    { name = "intersect (and number)"
    , pattern =
      [ Predicate isWithoutCents
      , regex "e"
      , Predicate isNatural
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney fd:
         _:
         Token Numeral NumeralData{TNumeral.value = c}:
         _) -> Just . Token AmountOfMoney $ withCents c fd
        _ -> Nothing
    }
  
  ruleDollar :: Rule
  ruleDollar = Rule
    { name = "$"
    , pattern =
      [ regex "dolares?"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly Dollar
    }
  
  ruleCent :: Rule
  ruleCent = Rule
    { name = "cent"
    , pattern =
      [ regex "centavos?"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly Cent
    }
  
  ruleIntersectXCents :: Rule
  ruleIntersectXCents = Rule
    { name = "intersect (X cents)"
    , pattern =
      [ Predicate isWithoutCents
      , Predicate isCents
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney fd:
         Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just c}:
         _) -> Just . Token AmountOfMoney $ withCents c fd
        _ -> Nothing
    }
  
  rulePounds :: Rule
  rulePounds = Rule
    { name = "£"
    , pattern =
      [ regex "libras?"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly Pound
    }
  
  ruleIntersectAndXCents :: Rule
  ruleIntersectAndXCents = Rule
    { name = "intersect (and X cents)"
    , pattern =
      [ Predicate isWithoutCents
      , regex "e"
      , Predicate isCents
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney fd:
         _:
         Token AmountOfMoney AmountOfMoneyData{TAmountOfMoney.value = Just c}:
         _) -> Just . Token AmountOfMoney $ withCents c fd
        _ -> Nothing
    }
  
  ruleIntersect :: Rule
  ruleIntersect = Rule
    { name = "intersect"
    , pattern =
      [ Predicate isWithoutCents
      , Predicate isNatural
      ]
    , prod = \tokens -> case tokens of
        (Token AmountOfMoney fd:
         Token Numeral NumeralData{TNumeral.value = c}:
         _) -> Just . Token AmountOfMoney $ withCents c fd
        _ -> Nothing
    }
  
  ruleBrl :: Rule
  ruleBrl = Rule
    { name = "BRL"
    , pattern =
      [ regex "reais|r\\$"
      ]
    , prod = \_ -> Just . Token AmountOfMoney $ currencyOnly BRL
    }
  
  rules :: [Rule]
  rules =
    [ ruleUnitAmount
    , ruleBrl
    , ruleCent
    , ruleDollar
    , ruleIntersect
    , ruleIntersectAndNumeral
    , ruleIntersectAndXCents
    , ruleIntersectXCents
    , rulePounds
    ]
  